
Lab 9. Use of DC motors and servo motors. The experimental robot. 

 

This lab work presents the use of DC motors, and the use of servo motors.  

 

1. DC motors 

 
Figure 1. DC Motor with 1:48 rotation rate reduction. 

 

The classic DC motors convert electrical energy into mechanical work. The rotation rate of a 

motor is proportional to the input voltage, and the rotation direction (clockwise or counter-

clockwise) depends on the polarity of the voltage (connecting the two wires of the motor to 

Vcc and GND, or vice versa). The motors have a gear box (reduction mechanism) with a 1:48 

reduction rate, which means that for a complete rotation of the external shaft 48 rotations of 

the motor shaft are needed. The reduction mechanism amplifies the force (torque), at the cost 

of speed. 

 

Due to the fact that the motors require a significant current intensity to produce movement, 

they cannot be connected directly to the output pins of a microcontroller. A separation 

between the command signals and the power circuit is required, and this is achieved by using 

H bridges. The H bridged are circuits based on four switches (usually transistors), S1, S2, S3 

and S4, as seen in figure 2. 

 
Figure 2. The H bridge: S1-S4 are the switches, and M is the motor. 

 



The name of “H bridge” comes from the shape of the power circuit. The upper left and lower 

right switches are usually connected to a common signal “A”, and the bottom left with the 

upper right switches are connected to a common signal “B”. Signals A and B must be 

exclusive, each one causing the motor to rotate in one direction. Activating both signals is 

forbidden, as it will short-circuit the power source. 

 

The two allowed states of the H bridge’s switches are shown in figure 3. 

 
Figure 3. The allowed states of the switches. 

 

By activating the switches S1 and S4 the motor will rotate in one direction, and by activating 

S2 and S3 the motor will rotate in the opposite direction.  

 

In this laboratory we will use the double bridge L298N Dual H-Bridge, capable of driving 

two DC motors in the same time. 

 
Figure 4. L298N Dual H-Bridge. 

Circuit specifications: 

 

- The motor driving voltage (the pin marked +12V) Vs: 5~35V; If we want to use the same 

source for powering the Arduino board, a voltage between 7 and 35V must be connected, 

to allow the integrated voltage regulator to generate 5 V on the +5V pin. 

- Maximum current for the motor driving circuit: 2A 

- Logic circuit voltage (the pin marked +5V) Vss: 5 - 7V (can be connected to Arduino + 

5V pin, for powering the microcontroller board) 

- Maximum current for the logic circuit: 36mA 



- Logic control signal levels: logic 0, －0.3≤Vin≤1.5V, logic 1, 2.3V≤Vin≤Vss 

- Maximum power: 20W 

 

The schematic is presented in figure 5: 

 

 
Figure 5. Schematic of the L298N driver circuit. 

 

For each motor there are three control pins. The first motor is connected to pins EnA, In1, si 

In2, and the second motor to pins EnB, In3 si In4. The En pins are connected to logic 1 using 

jumpers, and thus the user program will only control the In pins. The following combinations 

are available: 

 

In1 In2 Effect 

0 0 Motor 1 stopped (brake) 

0 1 Motor 1 on – forward 

1 0 Motor 1 on – reverse 

1 1 Motor 1 stopped (brake) 

 

In3 In4 Effect 

0 0 Motor 2 stopped (brake) 

0 1 Motor 2 on – forward 

1 0 Motor 2 on – reverse 

1 1 Motor 2 stop (brake) 

 

The motors can be driven simultaneously. Do not change the motor’s direction of rotation 

without stopping it first for several milliseconds.  

 



The rotation rate of a motor is controlled by the voltage applied to its power pins. Since a 

microcontroller can only generate 0 or 5 V voltage levels, for changing the rotation rate a 

PWM signal on pins In1, In2, In3 or In4 can be used. 

 

2. Servo motors 

 

Unlike the DC motors, which rotate continuously as long as they are connected to a power 

source, the servo motors are used to achieve partial, stable and controlled rotations, for small 

amplitude but high precision operations: locking mechanisms, positioning of sensors, 

gestures, etc. 

 

 
Figure 6. A servo motor. 

 

The servo motors have three wires, of different colors (depending on the manufacturer). The 

red color usually denotes Vcc (5V), while GND is usually black or brown. Besides the power 

wires, the command wire is usually yellow, orange or white. Figure 7 shows several color 

schemes.  

 

 
Figure 7. Color schemes used for servo motor wires. 

 

The servo motor will not (usually) execute a complete rotation, but will deviate from a 

neutral position with an angle controlled by the voltage applied to the signal pin. Using a 

PWM signal on this pin, we can control the motor’s angle of rotation.  

 



The simplest way of controlling the servo motors is by the use of the Servo library. With this 

library, up to 48 servos can be controlled by Arduino Mega. If more than 12 servos are used, 

the library will disable PWM on pins 11 and 12. On Arduino Uno, this library will disable 

PWM on pins 9 and 10, regardless of the number of servos used. 

 

The methods of the Servo class are the following: 

 

servo.attach(pin) / servo.attach(pin, min, max) – attaches the Servo object to pins  

● servo: a Servo class object  

● pin: number of digital pin for the servo control signal  

● min (optional): width of the pulse, in microseconds, for the minimum angle (0 

degrees) of the servo motor (implicitly 544)  

● max (optional): width of the pulse, in microseconds, for the maximum angle (180 

degrees) of the servo motor (implicitly 2400)  

 

servo.detach() – detaches the Servo object from the pin. 

boolean val servo.attached() – checks whether the Servo object is attached to a pin.  

 

servo.write (angle) – writes a value (0 .. 180) to the servo, controlling its motion: 

● For standard servos ⇒ sets the angle in degrees, causing the servo to orient in the 

specified position.  

● For continuously rotating servos ⇒ configures the rotation speed (0: maximum speed 

in one direction; 180: maximum speed in the opposite direction; ≈ 90: stopped) 

 

int val servo.read() – reads the current angle of the servo, set by the last call of write(). 

 

3. The experimental robot 

 

For the project activities, experimental robots have been assembled, with the following 

components: 

 

1. Arduino Uno compatible microcontroller board 

2. L298N Dual H-Bridge motor driver 

3. 2x DC Motor 

4. 1 Servo motor 

5. Battery case 4xAA (R6) 

6. 2 wheels connected to the motors, 1 balance wheel 

7. Acrylic support 

8. 2x Breadboard 

9. 1 sonar sensor (not connected) 

 



 
Figure 8. The experimental robot – top view – control elements. 

 

 
Figure 9. The experimental robot – bottom view – power circuit. 

 

The schematic of the connections between the elements of the robot is shown in figure 10. 

 



 
Figure 10. Electric schematic of the robot. 

 

Important features: 

- The motors can be powered by batteries (not optimal, according to the L298N 

specifications, but offers mobility), or by an external power supply connected to the 

Arduino jack power connector. The Vin pin of Arduino is connected internally to the jack 

connector, and also to the +12V pin of L298N. 

- Do not use an external power supply along with the batteries, as they will be exposes to 

the voltage of this source! 

- The +5V output of L298N is connected to the +5V power pin of the Arduino board. For 

saving pins, the ICSP connector has been used. 

- The control pins for the motors have been connected to Arduino pins 3 and 11 (Motor 1), 

and 5 and 6 (Motor 2), pins capable of PWM signal generation. 

- The servo control pin is connected to Arduino pin 8. 

 

Safety advice: 

- Do not let the robot start its motors while connected to the PC via the programming cable. 

The example program shown on the next page there is a built in warning time (the LED 

on pin 13 will blink, at first slower, then faster, before the motors are started). Program 

the robot, then disconnect the USB cable, and connect the external power supply, or the 

batteries. 

- Check the polarity of the power supply connector (the positive electrode must be in the 

center of the connector), and its set voltage (recommended 7.5V). 

- Do not let the robot start the motors while the wheels are on the table! It will move 

suddenly, and may cause breaking of cables, or it may fall from the table. Hold the robot 



in hand, or mount it on a stand to keep the wheels away from the table. When you want to 

test the robot’s motion, place it on the floor and leave enough space around it. 

 

4. Example code 

 

The following code is already programmed in the robot’s microcontroller. This code will test 

all motors, after the execution of the warning code. 

 

#include <Servo.h> 

 

// Pins of motor 1 

#define mpin00 5 

#define mpin01 6 

 

// Pins of motor 2 

#define mpin10 3 

#define mpin11 11 

 

Servo srv; 

 

void setup() { 

  // configuration of motor pins as output, initially 0 

  digitalWrite(mpin00, 0); 

  digitalWrite(mpin01, 0); 

  digitalWrite(mpin10, 0); 

  digitalWrite(mpin11, 0); 

 

  pinMode (mpin00, OUTPUT); 

  pinMode (mpin01, OUTPUT); 

  pinMode (mpin10, OUTPUT); 

  pinMode (mpin11, OUTPUT); 

 

  // LED pin 

  pinMode(13, OUTPUT);  

} 

 

// Function to control a motor 

// Input: pins m1 and m2, direction and speed 

void StartMotor (int m1, int m2, int forward, int speed) 

{ 

   

  if (speed==0) // stop 

  { 

    digitalWrite(m1, 0); 



    digitalWrite(m2, 0); 

  } 

  else  

  { 

    if (forward) 

    { 

      digitalWrite(m2, 0); 

      analogWrite (m1, speed); // use PWM 

    } 

    else 

    { 

      digitalWrite(m1, 0); 

      analogWrite(m2, speed); 

    } 

  } 

} 

 

// Safety function 

// Commands motors to stop, then delays 

void delayStopped(int ms) 

{ 

  StartMotor (mpin00, mpin01, 0, 0); 

  StartMotor (mpin10, mpin11, 0, 0); 

  delay(ms); 

} 

 

// Use of servo 

// Set three angles 

// When finished, the servo remains in the middle (90 degrees) 

void playWithServo(int pin) 

{ 

  srv.attach(pin); 

  srv.write(0); 

  delay(1000); 

  srv.write(180); 

  delay(1000); 

  srv.write(90); 

  delay(1000); 

  srv.detach(); 

 

} 

 

void loop() { 

 



 // Warning code 

 // Slow blink  

  for (int i=0; i<10; i++) 

  { 

    digitalWrite(13, 1); 

    delay(200); 

    digitalWrite(13, 0); 

    delay(200); 

  } 

 

  // Fast blink. Remove the USB cable!!!! 

  for (int i=0; i<10; i++) 

  { 

    digitalWrite(13, 1); 

    delay(100); 

    digitalWrite(13, 0); 

    delay(100); 

  } 

 

  digitalWrite(13, 1); 

 

  // Start the servo motor 

  playWithServo(8); 

 

  // Now start the DC motors 

  StartMotor (mpin00, mpin01, 0, 128); 

  StartMotor (mpin10, mpin11, 0, 128); 

   

  delay (500); // How long the motors are on 

  delayStopped(500); // How long the motors are off 

   

  StartMotor (mpin00, mpin01, 1, 128); 

  StartMotor (mpin10, mpin11, 1, 128); 

   

  delay (500); 

  delayStopped(500); 

 

  StartMotor (mpin00, mpin01, 0, 128); 

  StartMotor (mpin10, mpin11, 1, 128); 

   

  delay (500); 

  delayStopped(500); 

   

  StartMotor (mpin00, mpin01, 1, 128); 



  StartMotor (mpin10, mpin11, 0, 128); 

   

  delay (500); 

  delayStopped(500); 

}  

 

Project activity rules: 

 

1. Each team of students (1 or 2 students) will receive a robot. The lab assistant will write 

down in the attendance sheet the number of the robot for each student. In the following 

weeks, the students will use the same robot, if possible. 

2. Check the connections for compliance with the documentation. Run the test program, 

check for all motors to work correctly (each motor must perform rotations both ways). If 

defects are found, they will be identified and you will attempt to fix them. 

Step 2 is mandatory for each lab session! 

3. The teams will add functions to the robot, by changing the program and adding 

components. Avoid permanent changes, leave the robots in the same state as you found them. 

4. Each team will keep their code confidential. Each student is responsible for ensuring that 

no code remains on the workstation, and that their progress is securely saved. 

5. In the last lab session, the project will be evaluated based on functionality, complexity and 

originality. The students must hand out a short documentation (5-10 pages). 

 

Each damaged component will be replaced by the person that damaged it! 

 

Ideas for additional functions (may be cumulative): 

 

- Implement a precise control of the motors to ensure movement in a straight line. Implies 

measuring the rotation speed and tuning the PWM pulse to ensure equal rotation for both 

motors. 

- Detecting and avoiding obstacles. Implies using the sonar sensor along with the servo 

motor to determine distance and angle to obstacles, and to compute free space. 

- Mapping the environment. Implies detecting the obstacles and keeping track of the own 

movement, for placing the obstacles on the map. 

- Use of the EEPROM memory to remember a path previously traveled.  

- Remote control, using the mobile phone. Implies using the WiFi module. 

- Communication between robots using gestures.  

- Following a leader robot. 

 

It is possible that the additional functions may require additional hardware to be acquired. If 

the components are acquired by the students, they remain in their property. It is 

recommended that each team buys at least 4 alkaline batteries.  

 

Recommendation: you can take advantage of the features of the Mega boards without 

removing the UNO boards from the robot. Connect the boards by I2C, and use UNO only for 



motor control, and Mega for the other functions. Connect together the 5V and GND pins of 

the two boards, for sharing power. 

 

Additional references: 

 

1. Photoelectric sensor for using along with the code wheel, for measuring rotation speed. 

 

https://ardushop.ro/ro/home/146-senzor-de-intrerupere-

infrarosu.html?search_query=fotoelectric&results=1 

or 

https://www.robofun.ro/senzor-ir-break-beam-led-3mm 

 

2. Specifications of the sonar sensor for distance measurement 

 

https://cdn.sparkfun.com/datasheets/Sensors/Proximity/HCSR04.pdf 

 

3. Using the Sharp distance sensors 

 

http://www.robotshop.com/letsmakerobots/files/IR_SHARP.doc 

 

4. Use of the AVR built in EEPROM 

 

https://www.arduino.cc/en/Reference/EEPROM 

 

 


